
ABSTRACT  
Through the Output Delivery System (ODS), SAS® Software provides a means for creating an attractive, concise, 
and functional report without post-processing using other software products. Because ODS is an object-oriented (OO) 
technology, it provides the ability to move beyond template descriptions to complete output layout. This paper 
presents a combined text and graphics report based on an exploratory analysis of behavioral health cost risk using 
ODS LAYOUT and ODS REGION and writing to the PDF destination.  In addition, this paper provides information on 
additional options that can be used with both ODS LAYOUT and the PDF destination to assist users in the creation of 
customized presentation and production reports. 
 
The example in this paper was developed with version 9.1.3 of SAS executing on a Windows XP platform. SAS 
Version 9.1 is required for the ODS graphics extensions and was required for the graphics used in this paper. The 
SAS features represented in this paper are not platform-specific and can be adapted by both beginning and 
advanced SAS users.  

INTRODUCTION  
Health Management Corporation (HMC) is one of the nation's largest, most experienced providers of integrated care 
and total health solutions. Since 1983, HMC has offered members and clients comprehensive programs and services 
intended to empower members to take control of their health and see positive results. Through a population-based 
approach, HMC's disease management programs provide solutions for prevention, chronic condition support, lifestyle 
management and complex condition care. HMC's mission is to help improve the health and financial outcomes of its 
clients and members through innovative health solutions that consider every single member at his/her level of care. 
 
To fulfill this mission, HMC provides research, business intelligence, and program evaluation support to measure the 
effectiveness of the services provided, analyze potential for future products and add to general knowledge of the 
disease management arena. SAS is the major analytic software used by HMC and, as such, the graphics examples 
and report layout illustrated in this paper were produced with SAS software in conjunction with this work. SAS 
software provides all the tools necessary for data validation, analysis, and results reporting, eliminating the need to 
move output to other software for report formatting. 
  
 
ODS LAYOUT  
ODS LAYOUT provides the ability to more easily combine text and graphics in a report within SAS. Absolute layout is 
supported, with the page divided into regions. ODS LAYOUT in its initial release is limited to a single page; however, 
page-spanning features are under development. SAS code to use layout consists of ODS statements, global 
statements, and procedures, but instead of placing the output as streaming output, each piece is placed into the 
identified region. ODS dimension units can be in centimeters (cm), millimeters (mm), inches (in), printers pt (pt), or 
pixels (px).  
 
Absolute layout is ideal for reports that use a very precise format and provides flexibility within that context and has 
been available as experimental since SAS version 9.0. The exact position of each region is specified in the SAS 
code. For reports where the size and location of items do not need to be fixed, a relative layout option, gridded layout, 
is available as experimental beginning in SAS version 9.1. In gridded layout, the output area acts similar to a table 
where each cell in the table contains output from a single procedure or other item. This allows a series of rows and 
columns of different heights and widths and will also allow the expanding of space if the output does not fit in the 
“absolute” space. Both the absolute and gridded layout options allow regions to overlap.  
 
Use ABSOLUTE LAYOUT for: 
 

• Cover pages 
• Static data or reports  
• Pre-printed forms 
• Single page only 

 
Use GRIDDED LAYOUT for: 
 

• Regions that can be dynamically sized 

• Dynamic content 
• Alignment 
• Multiple page documents 

 
ODS LAYOUT is available for the SAS Printer family destinations:  PRINTER, PDF, PCL, and PS. 
Layout blocks always start with ODS LAYOUT START and end with ODS LAYOUT END. The syntax of ODS 
LAYOUT is as follows:  
 

ODS LAYOUT START options;  
ODS REGION options;  
SAS Code. . ; 
ODS LAYOUT START options;  
ODS REGION options;  
SAS Code. . ; 
ODS LAYOUT END;  

 
The ODS LAYOUT and REGION statements 
The ODS LAYOUT and REGION statements allow many options that provide control over placement of content.  
Availability varies based on whether the output uses gridded or absolute layout.  Options that can be used on the 
ODS LAYOUT statement include: 
 

• WIDTH= –  Width of the overall LAYOUT (absolute only).  
• HEIGHT=   – Height of the overall LAYOUT (absolute only). 
• COLUMN_GUTTER= – Space to leave between columns (gridded only). 
• ROW_GUTTER= – Space to leave between rows (gridded only) 

  
Options that can be used on the ODS REGION statement include: 
 

• X= -  Relative to upper left corner of LAYOUT space (absolute ). 
• Y= – Relative to upper left corner of LAYOUT space (absolute ). 
• WIDTH=  – Width of the REGION (absolute, gridded). 
• HEIGHT = – Height of the REGION (absolute, gridded). 
• COLUMN= – Column where the REGION should be placed (gridded only). 
• ROW= – Row where the REGION should be placed (gridded only). 
• COLUMN_SPAN= – Number of grid columns REGION should span (gridded only). 
• ROW_SPAN= – Number of grid rows REGION should span (gridded only). 

 
Options available beginning with SAS 9.2 
 

• ENTIREPAGE – LAYOUT should occupy entire page (absolute only). 
• CHAIN – Output that overflows REGION should flow into next designated REGION (absolute only). 
• CONTINUE – Duplicate LAYOUT and continue overflow contents into same REGION space in 

duplicated LAYOUT (absolute only). 
 
Because ODS LAYOUT is experimental, features, options, and syntax may change for the production version, 
especially for the gridded option.  
 
 
EXAMPLE USING ODS LAYOUT 
The example presented in this paper uses ABSOLUTE LAYOUT to create a one page project summary report of a 
cost risk analysis for behavioral health. A two-column report is created mixing graphics and text and includes a 
banner and a company logo. The ODS PDF destination is used. While ODS LAYOUT has been most often 
associated with PROC REPORT, PRINT, and TABULATE, the example here combines free-form text and images. 
The example illustrates use of separate regions for parts of the report as well as the combining of text and graphics 
within a single region. (See output at end of paper.)  Some of the other options illustrated in this paper are: 
 

• ODS escapechar – Sets separator character to introduce new syntax and separate options. 
• PREIMAGE – Add image before other elements. 
• POSTIMAGE – Add image after other elements. 
• FONT_FACE – Set font to Arial. 
• FONT_SIZE – Set font size to 12 for title, 9 for main text and 7.5 for figure captions. 
• FONT_WEIGHT – Set bold on for titles and figure captions. 
• JUST – Set justification (center for title – otherwise left). 

• STARTPAGE – While not currently implemented for absolute layout, this will eventually tell SAS that 
new procedures should not start new pages. 

• COLOR= – 
 
IMAGE SCALING 
Because this is a summary, the graphics had already been created earlier in the project and are pulled in from file. 
They could be created as part of the report program also.  An example could be as follows: 

 
ODS REGION height=3 in width=3 in x=4.1 in y=6 in; 
proc gmap map=maps.uscounty data=bhdatamap; 
   id state county; 
   choro pctbhgrp / legend=legend coutline=gray; 
run; 
quit; 

 
Images used in the example use the SCALEHEIGHT and SCALEWIDTH options to fit the available space. ODS 
LAYOUT images can be scaled with the following options: 
 

• NOSCALE –  The default.  
• SCALETOFIT – Make image conform to space available. 
• SCALEHEIGHT –  Scale the height to conform exactly to the space available and scale the width as 

necessary to match the height without distorting the image.   
• SCALEWIDTH –  Scale the width to conform exactly to the space available and scale the height as 

necessary to match the width without distorting the image.   
• SCALE – Scale the image to fit into the available space such that it doesn't distort the image.  
• TILE – Repeat the image (at its "regular" size) so as to fill up the available space.  
• X= Supply the width to use for the image scaling.   
• Y= Supply the height to use for the image scaling. 

 
 
SAS CODE TO CREATE EXAMPLE: 
The code below created the example at the end of this paper.  Text entries are truncated in this code for readability of 
the code. Text could also be read in from a file or stored in a macro.  
 
  ODS PDF file="r:\SESUG 2008\SESUG_bh.PDF" color=yes style=fancyprinter  

   STARTPAGE=NEVER; 
ODS escapechar="^"; 
ODS LAYOUT start width=8 in height=10 in; 
ODS REGION x=10 y=0 width=8 in height=2 in; 
   ODS PDF text="^S={preimage='R:\SESUG 2008\SESUG 2008.bmp^scaleheight'}";   
 
ODS REGION x=90 pct y=10.3 in width=1 in height=1 in; 
   ODS PDF text="^S={postimage='R:\SESUG 2008\hmc.bmp^scalewidth'}";   
 
ODS REGION x=10 y=1.3 in width=8 in height=1 in; 
   ODS PDF text= "^S={just=c font_weight=bold font_size=12pt 
  font_face=Arial}Exploration of Behaviorial Health Cost Risk"; 
 
ODS REGION x=0.3 in y=1.6 in width=3.5 in height=3.5 in; 
   ODS PDF text="^S={just=l font_size=9pt font_weight=bold 

font_face=Arial}Introduction";  
   ODS PDF text="^S={just=l font_size=9pt font_face=Arial}As a part of this 
  study, eight impactable behavioral health conditions were . . .";  
   ODS PDF text=" "; 
   ODS PDF text= "^S={just=l font_size=9pt font_weight=bold 
  font_face=Arial}Analysis";  
   ODS PDF text="^S={just=l font_size=9pt font_face=Arial}The first 
  exploratory procedure uses kernel density estimation (kde) to . . .";  
 
ODS REGION x=0.3 in y=3.95 in width=3.5 in height=8.1 in; 
   ODS PDF text="^S={preimage='R:\SESUG 2008\graph1.bmp^scalewidth'}";   
   ODS PDF text= "^S={just=l font_size=7.5pt font_weight=bold  
 font_face=Arial}Fig 1. Cost Impact of Behavioral Health."; 
   ODS PDF text=" "; 
   ODS PDF text= "^S={just=l font_size=9pt font_face=Arial}In this second 
  exploratory procedure, behavioral health admissions are plotted . . .";  

   ODS PDF text="^S={preimage='R:\SESUG 2008\graph2.bmp^scalewidth'}";   
   ODS PDF text= "^S={just=l font_size=7.5pt font_weight=bold  
 font_face=Arial}Fig 2. Behavioral Health Admissions Seasonality."; 
 
ODS REGION x=4.1 in y=1.6 in width=3.5 in height=8.7 in; 
   ODS PDF text= "^S={just=l font_size=9pt font_face=Arial}It is important to 
  not only know the behavioral health costs, but how much impact . . ."; 
   ODS PDF text="^S={preimage='R:\SESUG 2008\graph3.bmp^scalewidth'}";   
   ODS PDF text= "^S={just=l font_size=7.5pt font_weight=bold 
  font_face=Arial}Fig 3. Proportion of Total Cost."; 
   ODS PDF text=" ";  
   ODS PDF text= "^S={just=l font_size=9pt font_face=Arial}As part of the 
  cost question, it is also important to know whether there . . ."; 
   ODS PDF text="^S={preimage='R:\SESUG 2008\graph4.bmp^scalewidth'}";   
   ODS PDF text= "^S={just=l font_size=7.5pt font_weight=bold 
  font_face=Arial}Fig 4. Referrals."; 
   ODS PDF text=" "; 
   ODS PDF text= "^S={just=l font_size=9pt font_weight=bold 
  font_face=Arial}Conclusion";  
   ODS PDF text= "^S={just=l font_size=9pt font_face=Arial}Visual depictions 
  of data relationships assist analysts and other users in better . . .";  
run; 
ODS LAYOUT end; 
ODS _all_ close;  

 
 
FANCYPRINTER STYLE CODE: 
Although the paper changed text fonts to Arial, the base style that was used for the example was the fancyprinter 
style, listed below. Another formatting option would be to modify the style, changing the fonts and sizes within the 
style, rather than within the program.  
 

proc template;                                                                 
   define style Styles.fancyPrinter;                                           
      parent = styles.Printer;                                                 
      replace fonts /                                                          
         'TitleFont2' = ("Palatino, Arial, Times Roman",13pt,Bold Italic)      
         'TitleFont' = ("Palatino, Arial, Times Roman",13pt,Bold Italic)       
         'StrongFont' = ("Palatino, Book Antiqua, Times Roman",10pt,Bold)      
         'EmphasisFont' = ("Palatino, Book Antiqua, Times Roman",10pt,Bold     
         Italic)                                                               
         'FixedEmphasisFont' = ("Courier",10pt,Italic)                         
         'FixedStrongFont' = ("Courier",10pt,Bold)                             
         'FixedHeadingFont' = ("Courier",10pt,Bold)                            
         'BatchFixedFont' = ("SAS Monospace, Courier",8pt)                     
         'FixedFont' = ("Courier",10pt)                                        
         'headingEmphasisFont' = ("ITC Zaph Chancery, Palatino, Times Roman",  
         13pt,Bold Italic)                                                     
         'headingFont' = ("ITC Zaph Chancery, Palatino, Times Roman",12pt,     
         Italic)                                                               
         'docFont' = ("Palatino, Times Roman",10pt);                           
      replace GraphFonts /                                                     
         'GraphDataFont' = ("Times Roman",8pt)                                 
         'GraphValueFont' = ("Times Roman",10pt)                               
         'GraphLabelFont' = ("Times Roman",12pt,Bold)                          
         'GraphFootnoteFont' = ("Times Roman",12pt,Bold)                       
         'GraphTitleFont' = ("Times Roman",14pt,Bold);                         
      style Table from Output /                                                
         rules = ALL                                                           
         cellpadding = 4pt                                                     
         cellspacing = 0.25pt                                                  
         borderwidth = 0.75pt;                                                 
      style Graph from Graph                                                   
         "Graph attributes" /                                                  
         cellspacing = 0.25pt                                                  
         borderwidth = 0.75pt;                                                 
   end;  
run;         

ODS STYLE ATTRIBUTES 
Style attributes that can be used vary by the destination of the report.  The table below lists those attributes that are 
available for the PDF destination in SAS v 9.1.  Some additional features that are available in SAS v 9.2 include the 
ability to underline, overline, and line-through text. V 9.2 also adds the ability to password protect and/or encrypt the 
created files. 
 

ODS Style Attributes Available for PDF Destination 
PDF Style Attribute Description 
ASIS=  How to handle leading spaces and line breaks.  
BACKGROUND=  Color of the background of the table or graph  
BORDERCOLOR  Color of the border if the border is just one color.  
BORDERCOLORDARK  Darker color to use in a border that uses two colors to create 3-D effect. 
BORDERCOLORLIGHT  Lighter color to use in a border that uses two colors to create 3-D effect. 
BORDERWIDTH  Width of the border of the table.  
BOTTOMMARGIN=  Bottom margin for the document.  
CELLHEIGHT=  Height of the cell.  
CELLPADDING=  Amount of white space on each of the four sides of the text in a cell.  
CELLSPACING=  Thickness of the spacing between cells.  
CELLWIDTH=  Width of the cell.  
CONTRASTCOLOR=  Alternate colors for maps to be applied to the blocks in region areas in block maps. 
DROPSHADOW=  Whether to use a drop shadow effect for text in a graph.  
ENDCOLOR=  End color for a gradient effect in a graph.  
FLYOVER=  Text to show in a tool tip for the cell.  
FONT_FACE=  Specify font to use.  
FONT_SIZE=  Size of the font to use.  
FONT_STYLE=   Style of the font. 
FONT_WEIGHT=  Font weight.  
FONT_WIDTH=  Font width compared to the width of the usual design.  
FONT=   Specify a font definition. 
FOREGROUND=  Color of text or data items  
GRADIENT_DIRECTION=  Direction of gradient effect in either the X or Y axis direction.  
IMAGE=  Image to appear in the background. This image can be positioned or tiled.  
INDENT=  Set a numeric value to use as the indention depth.  
JUST=  Horizontal positioning for text or images. 
LEFTMARGIN=  Left margin for the document.  
LINESTYLE=  Line type to use in a graph (SAS/GRAPH line types).  
LINETHICKNESS=  Thickness (width) of a line that is part of a graph.  
LINKCOLOR=  Color for links that have not yet been visited.  
MARKERSIZE=  Size of the symbol used to represent data values.  
MARKERSYMBOL=  Symbol used to represent data values.  
NOBREAKSPACE=  How to handle space characters.  
OUTPUTHEIGHT=  Height for graphics in the document.  
OUTPUTWIDTH=  Width of the table or of the graph or line thickness.  
OVERHANGFACTOR=  Upper limit for extending the width of the column.  
POSTIMAGE=  Image to place after the table or cell.  
POSTTEXT=  Text to place after the cell or table.  
PREIMAGE=  Image to place before the table or cell.  
PRETEXT= Text to place before the cell or table.  
PROTECTSPECIALCHARACTERS= Determine how  <, >, and & are interpreted. 
RULES=  Types of rules to use in a table.  
STARTCOLOR=  Start color for a gradient effect in a graph.  
TOPMARGIN=  Top margin for the document.  
TRANSPARENCY=  Level of transparency for a graph.  

URL=  Specify a URL to link to.  
VJUST=  Vertical justification.  

CONCLUSION  
ODS LAYOUT provides the ability to easily mix text, graphics, and tables on the same page, arranging them in any 
position desired on that page. It provides the ability to run SAS code within the layout code or to use output created at 
an earlier date.  Additionally, a large number of formatting options are available to enhance the ODS output, 
especially for the PDF destination. This functionality in SAS allows the creation of production quality reports without 
post-processing.  This is especially valuable in the service industry where timely creation of reports, both scheduled 
and ad hoc, is a major part of securing and retaining business. 

SAS OUTPUT  

 


	ODS LAYOUT 
	ODS LAYOUT provides the ability to more easily combine text and graphics in a report within SAS. Absolute layout is supported, with the page divided into regions. ODS LAYOUT in its initial release is limited to a single page; however, page-spanning features are under development. SAS code to use layout consists of ODS statements, global statements, and procedures, but instead of placing the output as streaming output, each piece is placed into the identified region. ODS dimension units can be in centimeters (cm), millimeters (mm), inches (in), printers pt (pt), or pixels (px). 



